

PRÁCTICA 5: Sucesiones y Series

Sucesiones

En clase hemos visto cómo calcular límites de sucesiones, pero ¿cómo podemos calcular esos límites con *Maxima*?

Por ejemplo, la sucesión que tiene como término general $x_n = \left(1 + \frac{1}{n}\right)^n$.

```
(%i11) f(n):=(1+1/n)^n;
```

```
(%o11) f(n):=1+1/n
```

y se puede calcular el límite en $+\infty$ sin ninguna dificultad

```
(%i12) limit(f(n), n, inf);
```

```
(%o12) %e
```

Series numéricas

El estudio de la convergencia de una serie numérica no siempre es fácil y mucho menos el cálculo del valor de dicha suma. *Maxima* sabe calcular la suma de algunas series y siempre tenemos la posibilidad de aplicar algún criterio de convergencia.

<code>sum(expr, n, a, b)</code>	$\sum_{n=a}^b expr$
<code>sum(expr, n, a, b), simpsum</code>	simplifica la suma
<code>nusum (expr, x, a, plus)</code>	$\sum_{n=a}^b expr$

La orden genérica para calcular una suma, finita o infinita, es `sum`. Hace falta indicar qué sumamos, `sum` la variable que hace de índice y entre qué valores varía dicha variable.

```
(%i84) sum(i, i, 0, 100);
```

```
(%o84) 5050
```

Además hace falta indicarle que desarrolle y simplifique el sumatorio si queremos obtener un resultado más razonable. Esto se consigue con el operador `simpsum` que aparece escrito automáticamente si utilizamos el menú de *wxMaxima*.

```
(%i85) sum(i, i, 0, n);
```

```
(%o85)  $\sum_{i=0}^n i$ 
```

```
(%i86) sum(i, i, 0, n), simpsum;
```

```
(%o86)  $\frac{n^2+n}{2}$ 
```




Figura Ventana para calcular sumas

Práctica 5: Sucesiones y Series

La segunda posibilidad es usar la orden `nusum` que utiliza algunas reglas de sumación para obtener en algunos casos (series hipergeométricas principalmente) una expresión mejor del resultado. `nusum`

```
(%i87) nusum(i,i,0,n);
```

```
(%o87) 
$$\frac{n(n+1)}{2}$$

```

A cualquiera de estas dos órdenes se accede desde el menú **Análisis**→**Calcular suma**. Aparece la ventana de la Figura que tienes que rellenar con la expresión a sumar, variable y extremos. En la parte inferior de dicha ventana, puedes marcar **Simplificar** o **Nusum** para escoger entre las órdenes que hemos comentado.

El conjunto de series que sabe sumar *Maxima* no es demasiado grande, ya hemos dicho que es un problema difícil, pero tiene respuesta para algunas series clásicas. Por ejemplo,

```
(%i88) sum(1/n^2,n,1,inf),simpsum;
```

```
(%o88) 
$$\frac{\pi^2}{6}$$

```

o la serie armónica

```
(%i89) sum(1/n,n,1,inf),simpsum;
```

```
(%o89) 
$$\infty$$

```

que sabemos que no es convergente. Si intentamos sumar una progresión geométrica cualquiera nos pregunta por la razón y la suma sin mayores problemas

```
(%i90) sum(x^n,n,1,inf),simpsum;
```

```
Is |x|-1 positive, negative or zero? negative;
```

```
(%o90) 
$$\frac{x}{1-x}$$

```

En cambio no suma algunas otras como

```
(%i91) sum(1/n!,n,0,inf),simpsum;
```

```
(%o91) 
$$\sum_{n=0}^{\infty} \frac{1}{n!}$$

```

Para sumar Series de forma exacta podemos también aplicar la definición: calculamos la suma parcial n -sima S_n y luego tomamos límites.

Esto parece funcionar en las geométricas:

```
(%i1) a(n):=1/3^n;
```

```
(%o1) 
$$a(n) := \frac{1}{3^n}$$

```

```
(%i2) s(n):=nusum(a(n), n, 1, n);
```

```
(%o2) 
$$s(n) := nusum(a(n), n, 1, n)$$

```

```
(%i3) limit(s(n),n,inf);
```

```
(%o3) 
$$\frac{1}{2}$$

```

aritmético-geométricas:

```
(%i4) a(n):=(4*n-1)/2^n;
(%o4) a(n):=
$$\frac{4n-1}{2^n}$$


(%i5) s(n):=nusum(a(n), n, 1, n);
(%o5) s(n):=nusum(a(n), n, 1, n)

(%i6) limit(s(n), n, inf);
(%o6) 7
```

hipergeométricas y sumables por descomposición:

```
(%i7) a(n):=1/(n*(n-1)*(n-2));
(%o7) a(n):=
$$\frac{1}{n(n-1)(n-2)}$$


(%i10) s(n):=nusum(a(n), n, 3, n);
(%o10) s(n):=nusum(a(n), n, 3, n)

(%i11) limit(s(n), n, inf);
(%o11)  $\frac{1}{4}$ 
```

En otras series (como las telescópicas, con logaritmos o raíces) el método anterior no funciona.

```
(%i12) a(n):=1/sqrt(n)-1/sqrt(n+1);
(%o12) a(n):=
$$\frac{1}{\sqrt{n}} - \frac{1}{\sqrt{n+1}}$$


(%i13) s(n):=nusum(a(n), n, 1, n);
(%o13) s(n):=nusum(a(n), n, 1, n)

(%i14) limit(s(n), n, inf);
errexpl non-rational term ratio to nusum
```

```
(%o14) 
$$\lim_{n \rightarrow \infty} \sum_{n=1}^n \left( \frac{1}{\sqrt{n}} - \frac{1}{\sqrt{n+1}} \right)$$

```


En estos casos debemos conformarnos con calcular la Suma aproximada.

En el ejemplo que acabamos de plantear se sabe que la suma exacta es 1. Observemos lo lenta que es la convergencia.

```
(%i31) sap(n):=sum(a(n), n, 1, n), simpsum;
(%o31) sap(n):=
$$\sum_{n=1}^n a(n)$$


(%i34) l:makelist([n,sap(n)],n,100,2000,100),numer;
[[100,0.900496280979],[200,0.92946543841414],[300,0.94236095822958],[400,0.95006238305611],
[500,0.95532329483912],[600,0.9592091491776],[700,0.96223052126998],[800,0.96466673733312],
[900,0.96668516976736],[1000,0.96839302293795],[1100,0.96986256126605],[1200,0.97114450715876],
[1300,0.97227565134993],[1400,0.97328341574274],[1500,0.97418871335402],[1600,0.9750078088398],
[1700,0.97575356775156],[1800,0.97643631851869],[1900,0.97706446148702],[2000,0.97764490829951]]
```

```
(%i32) wxplot2d([discrete,1]);
```



Criterios de Convergencia

Como hemos dicho, calcular la suma de una serie es un problema difícil y lo que hemos hecho en clase es estudiar cuándo son convergentes sin calcular su suma. Aplicar el criterio de la raíz o del cociente para decidir la convergencia de una serie, reduce el problema al cálculo de un límite, que en el caso anterior es trivial, en el que *Maxima* sí puede ser útil. Por ejemplo, la serie $\sum_{n \geq 1} \frac{1}{n^n}$

```
(%i1) a(n):=1/n^n;
```

```
(%o1) a(n):=- $\frac{1}{n^n}$ 
```

```
(%i2) sum(1/n^n, n, 1, inf), simpsum;
```

```
(%o2) 
$$\sum_{n=1}^{\infty} \frac{1}{n^n}$$

```

no sabemos sumarla, pero el criterio del cociente

```
(%i3) limit(a(n+1)/a(n), n, inf);
```

```
(%o3) 0
```

nos dice que es convergente.

Desarrollo de Taylor

El desarrollo en serie de Taylor de una función parte de una idea sencilla: si el polinomio de Taylor de la función $\cos(x)$ se parece cada vez más a la función, ¿por qué parar?

```
(%i95) taylor(cos(x), x, 0, 4);
```

```
(%o95)  $1 - \frac{x^2}{2} + \frac{x^4}{24} + \dots$ 
```

¿Qué es eso de un polinomio con infinitos sumandos? Para eso acabamos de ver series de números. Ya sabemos en qué sentido podemos hablar de sumas infinitas.

niceindices(powerseries(f(x), x, a))	desarrollo en serie de potencias de la función f centrado en el punto a
--------------------------------------	--

Si en el menú **Análisis**→**Calcular serie**, marcamos **Serie de potencias** obtenemos el desarrollo en serie de potencias de la función

```
(%i96) niceindices(powerseries
        (cos(x), x, 0));
```

```
(%o96) 
$$\sum_{i=0}^{\infty} \frac{(-1)^i x^{2i}}{(2i)!}$$

```

La orden `powerseries` es la encargada del cálculo de dicho desarrollo y el comando que se antepone, `niceindices`, escribe la serie de Taylor de una función con índices más apropiados. Si no la utilizamos, simplemente el resultado aparece un poco más “enmarañado”:

```
(%i97) powerseries(cos(x), x, 0);
```

```
(%o97) 
$$\sum_{i16=0}^{\infty} \frac{(-1)^{i16} x^{2i16}}{(2i16)!}$$

```

pero es esencialmente el mismo. Funciona bastante bien para las funciones con las que hemos trabajado en clase:

```
(%i98) niceindices(powerseries(atan(x), x, 0));
```

```
(%o98) 
$$\sum_{i=0}^{\infty} \frac{(-1)^i x^{2i+1}}{2i+1}$$

```




Figura Ventana para el cálculo de la serie de Taylor

Ahora podemos dar una respuesta al problema de en dónde se parecen una función y su polinomio: la primera condición es que la serie de Taylor tiene que ser convergente. El conjunto de valores x en los que es convergente, hemos visto en clase que, depende del radio de convergencia de la serie. Vamos a calcularlo:

```
(%i5) a(i):=(-1)^i/(2*i+1);
```

```
(%o5) a(i):=
$$\frac{(-1)^i}{2i+1}$$

```

```
(%i6) radio:1/(limit(abs(a(i+1)/a(i)),i,inf));
```

```
(%o6) 1
```

Perfecto. Ya sabemos que la serie de Taylor de la función arcotangente converge para $x \in]-1, 1[$ y no es convergente en $\mathbb{R} \setminus [-1, 1]$ y el dibujo que vimos de su polinomio de Taylor tiene ahora sentido. Queda por saber qué ocurre para $x = \pm 1$. Eso te toca a tí.

Observación Se pueden realizar algunas operaciones con series de potencias. Por ejemplo, intenta derivarlas:

```
(%i101) f(x):=niceindices(powerseries(log(1+x^2), x, 0))$
```

```
(%i102) diff(f(x), x);
```

```
(%o102) 
$$-2 \sum_{i=1}^{\infty} (-1)^i x^{2i-1}$$

```

EJERCICIOS PROPUESTOS

1) Calcular los siguientes límites:

$$a) \lim_{n \rightarrow \infty} \left(\cos \frac{2}{n} + \operatorname{sen} \frac{2}{n} \right)^n$$

$$b) \lim_{n \rightarrow \infty} n \frac{\sqrt[n]{e} - e^{\operatorname{sen} \frac{1}{n}}}{1 - n \operatorname{sen} \frac{1}{n}}$$

$$c) \lim_{n \rightarrow \infty} (\sqrt{n+1} - \sqrt{n})\sqrt{n+1}$$

2) Calcular la suma de las siguientes series:

$$a) \sum_{n=1}^{\infty} \frac{1}{n^4}$$

$$b) \sum_{n=1}^{\infty} \frac{1}{a^n}$$

$$c) \sum_{n=1}^{\infty} \frac{6n+4}{3^n}$$

$$d) \sum_{n=1}^{\infty} \frac{n+1}{(n+3)(n+2)n}$$

3) Dada la serie telescópica de término general:

$$a(n) = \ln\left(1 - \frac{1}{n^2}\right)$$

cuya suma exacta es $-\ln 2$, calcular la suma aproximada y comprobar que la convergencia es muy lenta

4) Estudiar la convergencia de las siguientes series:

$$a) \sum_{n=1}^{\infty} \frac{2^n}{n}$$

$$b) \sum_{n=1}^{\infty} \left(\frac{n}{2n+1}\right)^n$$

$$c) \sum_{n=1}^{\infty} \left(\frac{n}{n+1}\right)^{n^2}$$

5) Hallar el desarrollo en serie de McLaurin y el radio de convergencia de las siguientes funciones:

$$a) y = \operatorname{sen} x$$

$$b) y = \cos x^2$$

$$c) y = \operatorname{sen}^2 x$$